Assignment 8:- Spherical Polar Co-ordinates

- Amogh G. Okade (EP22B020)

Introduction to spherical co-ordinates:-

Spherical polar co-ordinate system is a co-ordinate system for 3 dimensions where a point is defined by 3 variables:-

1. **ρ :-** It is the radial distance of a point from the origin.
   $\rho = \sqrt{x^2 + y^2 + z^2}$
2. **θ :-** The angle between the x-axis and the projection of the vector on the x-y plane.
   $\theta = \tan^{-1}\left(\frac{y}{x}\right)$
3. **ϕ :-** The angle between the vector and the z-axis.
   $\phi = \cos^{-1}\left(\frac{z}{\rho}\right)$

The transformation from Cartesian co-ordinates to spherical co-ordinates involve the following substitutions:-

$\mathbf{x} = \rho \sin\phi \cos\theta$; $\mathbf{y} = \rho \sin\phi \sin\theta$; $\mathbf{z} = \rho \cos\phi$

The Jacobian is,

$$\mathbf{J} = \begin{vmatrix} \sin\phi \cos\theta & \sin\phi \sin\theta & \cos\phi \\ \rho \cos\phi \cos\theta & \rho \cos\phi \sin\theta & -\rho \sin\phi \\ -\rho \sin\phi \sin\theta & \rho \sin\phi \cos\theta & 0 \end{vmatrix} = \rho^2 \sin\phi$$

Below is a 3D figure which will help understand better:-

(Functions used have been explained in comments)

In[1]:=

```
Graphics3D[{{Opacity[0.3], Orange, Specularity[White, 5], Sphere[]},
   (*Opacity makes the sphere more transparent*)
   {Thickness[0.004], Line[{{0, 0, 0}, {x, y, z}}]},
   (*The line along the vector of the point*)
   {Thick, Arrowheads[0.025], Arrow[
     {{{0, 0, 0}, {1.5, 0, 0}}, {{0, 0, 0}, {0, 0, 1.05}}, {{0, 0, 0}, {0, 1.15, 0}}}]},
   (*To specify the size of arrowheads; Arrow is a line with an arrowhead*)
   {Orange, Thickness[0.002], Dashing[{0.0075, 0.005}],
    (* 4 Dashed lines specified in the next 4 lines*)
    Line[{{{x, y, 0}, {x, y, z}},
      {{x, 0, 0}, {x, y, 0}},
      {{0, y, 0}, {x, y, 0}},
      {{x, y, 0}, {0, 0, 0}}}]},
   Text[#1, #2] & @@ # & /@  (*Writing texts for labelling using Slots*)
    {{"x", {1.5, 0, 0.075}}, {"y", {0, 1.15, 0.08}},
     {"z", {0, 0.075, 1.075}}, {"ρ", (1 / 2) {x + 0.01, y + 0.15, z + 0.05}},
     {Style["φ", Plain], (1 / 3) {x, y - 0.09, z + 0.095}},
     {Style["θ", Plain], {0.275, 0.15, 0}}, {"x", {x / 2 + 0.1, y + 0.1, 0}},
     {"y", {x + 0.175, y / 2 + 0.075, 0}}, {"z", {x, y + 0.05, z / 2}}}},
  ViewPoint → {2.84775, 1.72478, 0.604528},
  (*Chaning the point from which we view the figure*)
  Boxed → False, (*Removing the box around the sphere*)
  BaseStyle → Italic,
  PlotLabel → " Spherical Coordinates",
  LabelStyle → {FontFamily → "Helvetica", Plain, Bold}
 ] /. {x → Sqrt[3 / 2] / 2, y → Sqrt[3 / 2] / 2, z → 1 / 2}
```

Out[1]=



Solving a question using spherical polar co-ordinates

Q:- Find the volume of the solid that lies inside the sphere $x^2 + y^2 + z^2 = 1$ and outside the cylinder $x^2 + y^2 = 0.25$.

Ans:- We use the transformations -

$$x = \rho \sin\phi \cos\theta;\ y = \rho \sin\phi \sin\theta;\ z = \rho \cos\phi$$

We know that volume $= \iiint_S dx\,dy\,dz = \iiint_R |J|\,d\rho\,d\phi\,d\theta$

Now, we need to find the limits for ρ, ϕ and θ.

Dynamic Plot showing variation of ρ : -

In[2]:=

```
sphereANDaxes = Graphics3D[{{Opacity[0.075], Sphere[]},
     {Thin, Arrowheads[0.0125], Arrow[{{{0, 0, 0}, {2.75, 0, 0}},
        {{0, 0, 0}, {0, 0, 1.05}}, {{0, 0, 0}, {0, 1.05, 0}}}]},
     Text[#1, #2] & @@ # & /@
      {{"x", {2.95, 0, 0.075}}, {"y", {0, 1.075, 0.075}}, {"z", {0, 0.05, 1.075}}}},
    ViewPoint → {2.84775, 0.72478, 0.604528},
    Boxed → False] /. {x → Sqrt[3 / 2] / 2, y → Sqrt[3 / 2] / 2, z → 1 / 2};
cylinder = Graphics3D[{Opacity[0.3], Orange, Specularity[White, 5],
    Cylinder[{{0, 0, -√0.75}, {0, 0, √0.75}}, 0.5]}, Boxed → False];
t = 0;
f[t_] := (Pause[0.01]; t)
finPointRho = {};
zLineMove =
  Graphics3D[{{Thick, Darker[Green], Line[{{0, 0, 0.4875}, {0, 0.5, 0.51}}]},
    {Text[Style["r = 0.5", Darker[Green]], {0, 0.25, 0.575}]}}];
movingRho :=
 Graphics3D[{{Thick, Red, Line[{{0, 0, 0}, finPointRho[[Length[finPointRho]]]}]},
   {Text["ϕ", {0.01, 0.07, 0.19}]}}]
Dynamic[Show[sphereANDaxes, cylinder, movingRho, zLineMove]]
While[t ≤ 0.5, AppendTo[finPointRho, {0, f[t], f[t]}];
 t = t + 0.01]
```

Out[9]=



Note :- The red line has been moved along $\frac{x}{0} = \frac{y}{0.5} = \frac{z}{0.5}$ from (0, 0, 0) to (0, 0.5, 0.5). Every point has been parameterized as (0,t,t) and t has been varied from 0 to 0.5.

Explanation :- The point (0, 0.5, 0.5) lies on the cylinder and the variation of ρ considers all points with the same θ and ϕ.

Finding the limits for ρ :-

In[11]:=

```
phiTouchingCylinder2 =
  Graphics3D[{{Thick, Red, Line[{{0, 0, 0}, {x, √(0.25 - x^2) + 0.0025, z}}]},
     {Text["ϕ", {0.01, 0.07, 0.19}]},
     {Red, Text["ρ", {0.1, 0.375, 0.275}]}}] /. {x → -0.01, z → √0.3};
zLineMid =
  Graphics3D[{Thick, Darker[Green], Line[{{0, 0.01, z - 0.0125}, {0, 0.5, z + 0.015}}],
     {Text["r = 0.5", {0, 0.25, √0.4}]}}] /. z → √0.3;
textBox2 = Graphics[{{Opacity[0.001], Rectangle[{0, 0}, {1, 1}]},
    {Text[Style["We can see that, ", Black, 22.5], {0.5, 0.85}]},
    {Text[Style["sinϕ = r/ρ = 0.5/ρ", 14.5], {0.4, 0.525}]},
    {Text[Style["=> ρ = 1/(2 sinϕ)", 14.5], {0.4, 0.275}]}}];
GraphicsRow[{Show[sphereANDaxes, cylinder, phiTouchingCylinder2,
   zLineMid, ImageSize → Large], textBox2}, ImageSize → Large]
```

Out[14]=



We can see that,

$$\sin\phi = \frac{r}{\rho} = \frac{0.5}{\rho}$$

$$\Rightarrow \rho = \frac{1}{2\sin\phi}$$

$$\textbf{Therefore, } 0 \leqslant \rho \leqslant \frac{1}{2\sin\phi}$$

Note that the value of ρ depends on ϕ, as can be seen from the below dynamic plot of the variation of ϕ.

Dynamic Plot showing variation of ϕ :-

In[15]:=

```
n = √0.75;
f[z_] := (Pause[0.01]; z)
finPointPhi = {};
zLinelow = Graphics3D[{Thick, Green, Line[{{0, 0.0075, z - 0.01}, {0, 0.5, z + 0.0125}}],
    {Text["r = 0.5", {0, 0.25, - √0.635}]}}] /. z → - √0.75;
zEqualLine = Graphics3D[{{Thick, Blue, Line[{{0, 0, 0}, {0, 0, - √0.75}}]},
    {Text[Style[" √0.75 ", Blue, 9], {0, -0.175, - √0.75/1.65}]}}];
movePhiTouching :=
 Graphics3D[{{Thick, Red, Line[{{0, 0, 0}, finPointPhi[[Length[finPointPhi]]]}]},
   {Text[Style["ϕ", 5 * (2 - n/1.5)], {0.01, 0.15 - Abs[n / 12], Sin[n]/8 + 0.15}]}}]
Dynamic[Show[sphereANDaxes, cylinder, movePhiTouching, zLinelow, zEqualLine]]
While[n ≥ - √0.70, AppendTo[finPointPhi, {0, 0.5, f[n]}];
 n = n - √0.75/100]
```

Out[21]=



Note:- To move the line, end points having co-ordinates (0,0.5,n) have been used.

Explanation:- The end points of the line lie on the cylinder $x^2 + y^2 = 0.25$. Thus, values of x and y have been fixed (here, 0 and 0.5 respectively) and the value of z has been varied along the side of the cylinder.

Finding the limits for ϕ :-

In[23]:=
```
phiTouchingCylinder = Graphics3D[{{Thick, Red, Line[{{0, 0, 0}, {x, Sqrt[0.25 - x^2], z}}]},
    {Text[Style["ϕmin", 10], {0.01, 0.11, 0.115}]},
    {Red, Text["ρ = 1", {0.1, 0.375, 0.4}]}}] /. {x → 0, z → Sqrt[0.75]};
zLine = Graphics3D[{Thick, Green, Line[{{0, 0.0075, z}, {0, 0.5, z + 0.0125}}],
    {Text["r = 0.5", {0, 0.25, Sqrt[0.835]}]}}] /. z → Sqrt[0.75];
textBox1 = Graphics[{{Opacity[0.001], Rectangle[{0, 0}, {1, 1}]},
   {Text[Style["As we can see, ", Black, 30], {0.45, 0.85}]},
   {Text[Style["sinϕmin = r/ρ = 0.5/1 = 1/2", 20], {0.4, 0.6}]},
   {Text[Style["=> ϕmin = π/6", 20], {0.4, 0.425}]},
   {Text[Style["Similarly, ϕmax = π - π/6 = 5π/6", 20], {0.45, 0.225}]}}];
GraphicsRow[
 {Show[sphereANDaxes, cylinder, phiTouchingCylinder, zLine, ImageSize → Large],
  textBox1}, ImageSize → Full]
```

Out[26]=



As we can see,

$$\text{sin}\phi_{min} = \frac{r}{\rho} = \frac{0.5}{1} = \frac{1}{2}$$

$$=> \phi_{min} = \frac{\pi}{6}$$

$$\text{ilarly, } \phi_{max} = \pi - \frac{\pi}{6} = \frac{5\pi}{6}$$

Therefore, $\frac{\pi}{6} \leq \phi \leq \frac{5\pi}{6}$

Dynamic Plot showing variation of θ : -

In[27]:=

```
x = 0;
f[x_] := (Pause[0.01]; x)
finPointThe = {};
movingThe :=
 Graphics3D[{{Thick, Red, Line[{{0, 0, 0}, finPointThe[[Length[finPointThe]]]}]}}]
Dynamic[Show[sphereANDaxes, cylinder, zLineMove, movingThe]]
mult = 1;
c = 0;
While[c ≠ 200,
  AppendTo[finPointThe, {f[x], mult * Sqrt[0.25 - f[x]^2], 0.5}];
 If[x == 0.5, mult = -1];
 If[x == -0.5, mult = 1];
 c = c + 1;
 x = x + 0.01 * mult]
```

Out[31]=



Note :- The move the red line, end points of the form $\{x, m * \sqrt{0.25 - x^2}, 0.5\}$ have been used, where m=1 in the 1st and 4th quadrants; and m=-1 in the 2nd and 3rd quadrants. x has been incremented by 0.01*m, which makes x go from 0 -> 0.5 -> 0 -> -0.5 -> 0.

Explanation :- The end points of the line have been moved along the circle $x^2 + y^2 = 0.25$; keeping z constant at z = 0.5.

Finding the limits for θ :-

In[35]:=

```
notMove = Graphics[{{Thick, Lighter[Red], Circle[{0, 0}, 0.5]}, {Thick, Darker[Cyan],
      Line[{{0, 0}, {0.5, 0}}]}, {Thick, Green, Line[{{0, 0}, {0.35, Sqrt[0.25 - 0.35^2]}}]},
     {Text[Style["θ", 15], {0.075, 0.035}]}}, Axes → True, ImageSize → Medium];
y = 0;
f[y_] := (Pause[0.01]; y)
finPointThe2 = {};
move := Graphics[{{Thick, Lighter[Red], Circle[{0, 0}, 0.5]},
    {Thick, Darker[Cyan], Line[{{0, 0}, {0.5, 0}}]},
    {Thick, Green, Line[{{0, 0}, finPointThe2[[Length[finPointThe2]]]}]}},
   Axes → True, PlotLabel → Style["Dynamic Variation of θ in 2D", 15]]
mult = -1;
c = 0;
GraphicsRow[{notMove, Dynamic[move]}]
While[c ≠ 201,
  AppendTo[finPointThe2, {-mult * Sqrt[0.25 - f[y]^2], f[y]}];
 If[y == -0.5, mult = 1];
 If[y == 0.5, mult = -1];
 c = c + 1;
 y = y + 0.01 * mult]
```

Out[42]=



Therefore, $0 \leqslant \theta \leqslant 2\pi$

Solving the question :-

$$\text{Volume} = \iiint_R |J|\, d\rho\, d\phi\, d\theta = \int_0^{2\pi} \int_{\frac{\pi}{6}}^{\frac{5\pi}{6}} \int_0^{\frac{1}{2\sin\phi}} \rho^2 \sin\phi\, d\rho\, d\phi\, d\theta$$

$$\text{Volume} = \frac{\pi}{2\sqrt{3}}$$

In[44]:=
```
Integrate[ρ^2 * Sin[ϕ], {θ, 0, 2 π}, {ϕ, π/6, 5 π/6}, {ρ, 0, 1/(2 Sin[ϕ])}]
```

Out[44]=

$$\frac{\pi}{2\sqrt{3}}$$

Extra Points

Explaining opacity :-

In[45]:=
```
a = Graphics3D[{Orange, Sphere[]}, PlotLabel → "No opacity"];
b = Graphics3D[{Opacity[0.3], Orange, Sphere[]}, PlotLabel → "Opacity=0.3"];
GraphicsRow[{a, b}, ImageSize → Medium]
```

Out[47]=



Explaining specularity :-

In[48]:=
```
a = Graphics3D[{Orange, Sphere[]}, PlotLabel → "No specularity"];
b =
  Graphics3D[{Specularity[White, 5], Orange, Sphere[]}, PlotLabel → "Specularity=5"];
c = Graphics3D[
   {Specularity[White, 15], Orange, Sphere[]}, PlotLabel → "Specularity=15"];
GraphicsRow[{a, b, c}, ImageSize → Medium]
```

Out[51]=



Explaining Arrowheads :-

In[52]:=
```
a = Graphics[{Line[{{0, 0}, {1, 1}}]}, PlotLabel → "Using Line"];
b = Graphics[{Arrowheads[0.1], Arrow[{{0, 0}, {1, 1}}]}, PlotLabel → "Using Arrow,
 Arrowhead size=0.1"];
c = Graphics[{Arrowheads[0.5], Arrow[{{0, 0}, {1, 1}}]}, PlotLabel → "Using Arrow,
 Arrowhead size=0.5"];
GraphicsRow[{a, b, c}, ImageSize → Large]
```

Out[55]=



Explaining Dashing :-

The first argument of dashing is for the length of the line that appears and the second argument is for the length of the gap.

In[56]:=
```
a = Graphics[{Orange, Dashing[{0.03, 0.03}], Line[{{{0, 0}, {1, 1}}}]},
   PlotLabel → "Line Length=0.03;
Gap Length=0.03"];
b = Graphics[{Orange, Dashing[{0.075, 0.05}], Line[{{{0, 0}, {1, 1}}}]},
   PlotLabel → "Line Length=0.075;
Gap Length=0.05"];
GraphicsRow[{a, b}, ImageSize → Medium]
```

Out[58]=



Explaining ViewPoint :-

In[59]:=
```
a = Graphics3D[{Sphere[]}, ViewPoint → {5, 5, 5}, PlotLabel → "ViewPoint is {5,5,5}"];
b = Graphics3D[{Sphere[]}, ViewPoint → {0, 5, 2}, PlotLabel → "ViewPoint is {0,5,2}"];
GraphicsRow[{a, b}, ImageSize → Medium]
```

Out[61]=



Comments

Thus, we can conclude that spherical polar co-ordinates can be used to easily visualize a lot of co-ordinate systems, especially those which involve spheres.